This worksheet deals with the linearized Euler equations for an incompressible fluid. See J.-F. Pommaret, *Partial Differential Equations and Group Theory: New Perspectives for Applications*, Kluwer, 1994, p. 671.

```
> with(Ore_algebra):
> with(OreModules):
```

We define the Ore algebra *Alg* to be the Weyl algebra, where *x1*, *x2*, *x3* are the spatial variables and *x4* is the time variable, and *Di* acts as differential operator w.r.t. *x*i:

```
> Alg := DefineOreAlgebra(diff=[D1,x1], diff=[D2,x2], diff=[D3,x3], diff=[D4,x4],
> polynom=[x1,x2,x3,x4]):
```

The linearized Euler equations are given by the following matrix. The first row stands for the divergence of the speed of the fluid and the three remaining rows give the sum of the partial derivative of the speed w.r.t. time and the gradient of the pressure of the fluid.

```
> R := evalm([[D1,D2,D3,0], [D4,0,0,D1], [0,D4,0,D2], [0,0,D4,D3]]);
```

$$R := \begin{bmatrix} \mathrm{D1} & \mathrm{D2} & \mathrm{D3} & 0 \\ \mathrm{D4} & 0 & 0 & \mathrm{D1} \\ 0 & \mathrm{D4} & 0 & \mathrm{D2} \\ 0 & 0 & \mathrm{D4} & \mathrm{D3} \end{bmatrix}$$

The linearized Euler equations are defined by:

```
> ApplyMatrix(R, [seq(v[i](x1,x2,x3,x4),i=1..3),p(x1,x2,x3,x4)], Alg)=
> evalm([[0]$4]);
```

$$\begin{bmatrix} (\frac{\partial}{\partial x1} v_1(x1, x2, x3, x4)) + (\frac{\partial}{\partial x2} v_2(x1, x2, x3, x4)) + (\frac{\partial}{\partial x3} v_3(x1, x2, x3, x4)) \\ (\frac{\partial}{\partial x4} v_1(x1, x2, x3, x4)) + (\frac{\partial}{\partial x1} \mathrm{p}(x1, x2, x3, x4)) \\ (\frac{\partial}{\partial x4} v_2(x1, x2, x3, x4)) + (\frac{\partial}{\partial x2} \mathrm{p}(x1, x2, x3, x4)) \\ (\frac{\partial}{\partial x4} v_3(x1, x2, x3, x4)) + (\frac{\partial}{\partial x3} \mathrm{p}(x1, x2, x3, x4)) \end{bmatrix} = \begin{bmatrix} 0 \\ 0 \\ 0 \\ 0 \end{bmatrix}$$

We compute the formal adjoint of R:

```
> R_adj := Involution(R, Alg);
```

$$R\_adj := \begin{bmatrix} -\mathrm{D1} & -\mathrm{D4} & 0 & 0 \\ -\mathrm{D2} & 0 & -\mathrm{D4} & 0 \\ -\mathrm{D3} & 0 & 0 & -\mathrm{D4} \\ 0 & -\mathrm{D1} & -\mathrm{D2} & -\mathrm{D3} \end{bmatrix}$$

```
> Ext1 := Exti(R_adj, Alg, 1);
```

$$Ext1 := \left[\begin{bmatrix} \mathrm{D2}^2\,\mathrm{D4} + \mathrm{D4}\,\mathrm{D1}^2 + \mathrm{D3}^2\,\mathrm{D4}\,, 0\,, 0\,, 0 \\ 0\,, \mathrm{D2}^2\,\mathrm{D4} + \mathrm{D4}\,\mathrm{D1}^2 + \mathrm{D3}^2\,\mathrm{D4}\,, 0\,, 0 \\ 0\,, 0\,, \mathrm{D2}^2\,\mathrm{D4} + \mathrm{D4}\,\mathrm{D1}^2 + \mathrm{D3}^2\,\mathrm{D4}\,, 0 \\ 0\,, 0\,, 0\,, \mathrm{D1}^2 + \mathrm{D2}^2 + \mathrm{D3}^2 \end{bmatrix}, \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}, \mathrm{SURJ}(4)\right]$$

We see that the *Alg*-module M which is associated with the system of linearized Euler equations is torsion, i.e., every element of M is a torsion element, because the generating set for $\mathrm{t}(M)$ which is given by the rows of *Ext1*[2] generates M. On the diagonal of *Ext1*[1] we find the operators that kill the corresponding torsion elements, more precisely, the entry *Di* in the *i*th column of *Ext1*[1] kills the *i*th torsion element θ_i, i=1, ..., 4, i.e., *Di* $\theta_i = 0$ in M. This information can be displayed in a more familiar way by using *TorsionElements* (here [*vx*, *vy*, *vz*] is the speed and *p* the pressure of the fluid):

```
> TorsionElements(R,
> [vx(x1,x2,x3,x4),vy(x1,x2,x3,x4),vz(x1,x2,x3,x4),p(x1,x2,x3,x4)], Alg);
```

$$\left[\left[\begin{array}{c} (\frac{\partial^3}{\partial x4\,\partial x1^2}\,\%4) + (\frac{\partial^3}{\partial x4\,\partial x3^2}\,\%4) + (\frac{\partial^3}{\partial x4\,\partial x2^2}\,\%4) = 0 \\ (\frac{\partial^3}{\partial x4\,\partial x1^2}\,\%3) + (\frac{\partial^3}{\partial x4\,\partial x3^2}\,\%3) + (\frac{\partial^3}{\partial x4\,\partial x2^2}\,\%3) = 0 \\ (\frac{\partial^3}{\partial x4\,\partial x1^2}\,\%2) + (\frac{\partial^3}{\partial x4\,\partial x3^2}\,\%2) + (\frac{\partial^3}{\partial x4\,\partial x2^2}\,\%2) = 0 \\ (\frac{\partial^2}{\partial x2^2}\,\%1) + (\frac{\partial^2}{\partial x3^2}\,\%1) + (\frac{\partial^2}{\partial x1^2}\,\%1) = 0 \end{array}\right], \left[\begin{array}{c} \%4 = \mathrm{vx}(x1,\, x2,\, x3,\, x4) \\ \%3 = \mathrm{vy}(x1,\, x2,\, x3,\, x4) \\ \%2 = \mathrm{vz}(x1,\, x2,\, x3,\, x4) \\ \%1 = \mathrm{p}(x1,\, x2,\, x3,\, x4) \end{array}\right]\right]$$

$$\%1 := \theta_4(x1,\, x2,\, x3,\, x4)$$
$$\%2 := \theta_3(x1,\, x2,\, x3,\, x4)$$
$$\%3 := \theta_2(x1,\, x2,\, x3,\, x4)$$
$$\%4 := \theta_1(x1,\, x2,\, x3,\, x4)$$

Summarizing, we note that *Exti* provides a way to find the relations that every element of the system satisfies on its own, i.e., the relations of the autonomous elements.

We now consider the Boussinesq stationary system for the Benard problem as it is described in J.-F. Pommaret, *Partial Differential Equations and Group Theory: New Perspectives for Applications*, Kluwer, 1994, p. 671-672.

First of all, let us define the Laplacian operator.

```
> Laplacian := D1^2+D2^2+D3^2;
```

$$\mathit{Laplacian} := \mathrm{D2}^2 + \mathrm{D1}^2 + \mathrm{D3}^2$$

Then, the Boussinesq stationary system for the Benard problem is defined by means of the following matrix:

```
> R2 := evalm([[D1,D2,D3,0,0],[Laplacian,0,0,-D1,0],[0,Laplacian,0,-D2,0],
> [0,0,Laplacian,-D3,g],[0,0,g,0,Laplacian]]);
```

$$\mathit{R2} := \left[\begin{array}{ccccc} \mathrm{D1} & \mathrm{D2} & \mathrm{D3} & 0 & 0 \\ \%1 & 0 & 0 & -\mathrm{D1} & 0 \\ 0 & \%1 & 0 & -\mathrm{D2} & 0 \\ 0 & 0 & \%1 & -\mathrm{D3} & g \\ 0 & 0 & g & 0 & \%1 \end{array}\right]$$

$$\%1 := \mathrm{D2}^2 + \mathrm{D1}^2 + \mathrm{D3}^2$$

If we denote by ($v[1]$, $v[2]$, $v[3]$) the speed of the viscous fluid, π and θ the perturbations of the pressure and temperature around the steady state, then the system is defined by:

```
> ApplyMatrix(R2, [seq(v[i](x1,x2,x3),i=1..3),pi(x1,x2,x3), theta(x1,x2,x3)], Alg)
> =evalm([[0]$5]);
```

$$\left[\left(\frac{\partial}{\partial x1}\, v_1(x1,\, x2,\, x3)\right) + \left(\frac{\partial}{\partial x2}\, v_2(x1,\, x2,\, x3)\right) + \left(\frac{\partial}{\partial x3}\, v_3(x1,\, x2,\, x3)\right)\right]$$

$$\left[\left(\frac{\partial^2}{\partial x2^2}\, v_1(x1,\, x2,\, x3)\right) + \left(\frac{\partial^2}{\partial x3^2}\, v_1(x1,\, x2,\, x3)\right) + \left(\frac{\partial^2}{\partial x1^2}\, v_1(x1,\, x2,\, x3)\right) - \left(\frac{\partial}{\partial x1}\, \pi(x1,\, x2,\, x3)\right)\right]$$

$$\left[\left(\frac{\partial^2}{\partial x2^2}\, v_2(x1,\, x2,\, x3)\right) + \left(\frac{\partial^2}{\partial x3^2}\, v_2(x1,\, x2,\, x3)\right) + \left(\frac{\partial^2}{\partial x1^2}\, v_2(x1,\, x2,\, x3)\right) - \left(\frac{\partial}{\partial x2}\, \pi(x1,\, x2,\, x3)\right)\right]$$

$$\left[\left(\frac{\partial^2}{\partial x2^2}\, v_3(x1,\, x2,\, x3)\right) + \left(\frac{\partial^2}{\partial x3^2}\, v_3(x1,\, x2,\, x3)\right) + \left(\frac{\partial^2}{\partial x1^2}\, v_3(x1,\, x2,\, x3)\right) - \left(\frac{\partial}{\partial x3}\, \pi(x1,\, x2,\, x3)\right) + g\, \theta(x1,\, x2,\, x3)\right]$$

$$\left[g\, v_3(x1,\, x2,\, x3) + \left(\frac{\partial^2}{\partial x2^2}\, \theta(x1,\, x2,\, x3)\right) + \left(\frac{\partial^2}{\partial x3^2}\, \theta(x1,\, x2,\, x3)\right) + \left(\frac{\partial^2}{\partial x1^2}\, \theta(x1,\, x2,\, x3)\right)\right]$$

$$= \begin{bmatrix} 0 \\ 0 \\ 0 \\ 0 \\ 0 \end{bmatrix}$$

Let us compute the rank of the *Alg*-module associated with *R2*.

```
> OreRank(R2, Alg);
```

$$0$$

Therefore, the *Alg*-module associated with *R2* is torsion, meaning that the system variables $v[\mathrm{i}]$, π and θ satisfy partial differential equations by themselves. Let us decouple the system variables by computing the torsion module using *OreModules*. We first need to define the formal adjoint *R2_ adj* of *R2*.

```
> R2_adj := Involution(R2, Alg);
```

$$R2\_adj := \begin{bmatrix} -\mathrm{D1} & \%1 & 0 & 0 & 0 \\ -\mathrm{D2} & 0 & \%1 & 0 & 0 \\ -\mathrm{D3} & 0 & 0 & \%1 & g \\ 0 & \mathrm{D1} & \mathrm{D2} & \mathrm{D3} & 0 \\ 0 & 0 & 0 & g & \%1 \end{bmatrix}$$

$$\%1 := \mathrm{D2}^2 + \mathrm{D1}^2 + \mathrm{D3}^2$$

Let us compute ext^1 of the *Alg*-module associated with *R2_ adj*.

```
> st := time(): ext1 := Exti(R2_adj, Alg, 1); time()-st;
```

$$ext1 := \Big[$$

$$[D1^8 - 2\,g^2\,D1^2\,D2^2 - g^2\,D1^4 - g^2\,D2^4 - g^2\,D3^2\,D1^2 - g^2\,D3^2\,D2^2 + 12\,D1^4\,D3^2\,D2^2 + 6\,D1^4\,D3^4 + 6\,D1^4\,D2^4 + 4\,D1^6\,D3^2 + 4\,D1^6\,D2^2 + D3^8 + 4\,D1^2\,D2^6 + 12\,D1^2\,D2^2\,D3^4 + 12\,D1^2\,D2^4\,D3^2 + 6\,D2^4\,D3^4 + 4\,D2^6\,D3^2 + 4\,D1^2\,D3^6 + 4\,D2^2\,D3^6 + D2^8,\,0\,,\,0\,,\,0\,,\,0]$$

$$[0\,,D1^8 - 2\,g^2\,D1^2\,D2^2 - g^2\,D1^4 - g^2\,D2^4 - g^2\,D3^2\,D1^2 - g^2\,D3^2\,D2^2 + 12\,D1^4\,D3^2\,D2^2 + 6\,D1^4\,D3^4 + 6\,D1^4\,D2^4 + 4\,D1^6\,D3^2 + 4\,D1^6\,D2^2 + D3^8 + 4\,D1^2\,D2^6 + 12\,D1^2\,D2^2\,D3^4 + 12\,D1^2\,D2^4\,D3^2 + 6\,D2^4\,D3^4 + 4\,D2^6\,D3^2 + 4\,D1^2\,D3^6 + 4\,D2^2\,D3^6 + D2^8,\,0\,,\,0\,,\,0]$$

$$[0\,,\,0\,,\,\%1\,,\,0\,,\,0]$$

$$[0\,,\,0\,,\,0\,,\,\%1\,,\,0]$$

$$[0\,,\,0\,,\,0\,,\,0\,,\,\%1],\ \left[\begin{array}{ccccc} 1 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 0 & 1 \end{array}\right],\ \mathrm{SURJ}(5)\Big]$$

$$\%1 := D1^6 + 3\,D1^4\,D2^2 + 3\,D1^4\,D3^2 + 6\,D1^2\,D3^2\,D2^2 + 3\,D1^2\,D3^4 + 3\,D1^2\,D2^4 - D1^2\,g^2 + 3\,D2^4\,D3^2 + 3\,D2^2\,D3^4 + D2^6 - g^2\,D2^2 + D3^6$$

2.220

We find againthat the *Alg*-module associated with the Boussinesq system is torsion. Moreover, the differential operators on the diagonal of the first matrix *ext1*[1] are the operators that kill respectively $v[1]$, $v[2]$, $v[3]$, π and θ.

More precisely, the first component of the speed $v[1]$ satisfies the equation $P1\ v[1] = 0$, where $P1$ is defined by:

```
> P1 := collect(ext1[1][1,1], g, distributed, factor);
```

$$P1 := (D3^2 + D1^2 + D2^2)^4 - (D3^2 + D1^2 + D2^2)\,(D1^2 + D2^2)\,g^2$$

The second component of the speed $v[2]$ satisfies the equation $P2\,v[2] = 0$, where $P[2]$ is defined by:

```
> P2 := collect(ext1[1][2,2], g, distributed, factor);
```

$$P2 := (D3^2 + D1^2 + D2^2)^4 - (D3^2 + D1^2 + D2^2)\,(D1^2 + D2^2)\,g^2$$

The third component of the speed $v[3]$ satisfies the equation $P3\,v[3] = 0$, where $P[3]$ is defined by:

```
> P3 := collect(ext1[1][3,3], g, distributed, factor);
```

$$P3 := (D3^2 + D1^2 + D2^2)^3 + (-D1^2 - D2^2)\,g^2$$

The perturbation of the pressure π satisfy the equation $P4\,\pi = 0$, where $P4$ is defined by:

```
> P4 := collect(ext1[1][4,4], g, distributed, factor);
```

$$P4 := (D3^2 + D1^2 + D2^2)^3 + (-D1^2 - D2^2)\,g^2$$

Finally, the perturbation of the temperature θ satisfies $P5\,\theta = 0$, where $P5$ is defined by:

```
> P5 := collect(ext1[1][5,5], g, distributed, factor);
```

$$P5 := (\mathrm{D3}^2 + \mathrm{D1}^2 + \mathrm{D2}^2)^3 + (-\mathrm{D1}^2 - \mathrm{D2}^2)\, g^2$$